

**MODEL 35B
GENERAL PURPOSE DIFFERENTIAL AMPLIFIER**

LAWSON LABS, INC.
3217 Phoenixville Pike
Malvern, PA 19355

610 725-8800
or
800 321-5355

Rev. D

TABLE OF CONTENTS

INTRODUCTION	Page 2
1. INTERCONNECTIONS	Page 2
2. CONNECTING INPUTS	Page 3
3. OPERATION	Page 5
4. USING MULTIPLE MODEL 35Bs	Page 8
5. TROUBLESHOOTING	Page 10
6. INPUT CODE SUMMARY	Page 11
7. INTERNAL ADJUSTMENT & CALIBRATION	Page 11
8. SPECIFICATIONS	Page 12
WARRANTY	Page 13

INTRODUCTION

The Lawson Labs Model 35B Differential Amplifier, in combination with any Lawson Labs precision A/D card, allows a personal computer to be used as a multi-channel millivoltmeter. Despite the low per channel cost, the Model 35B delivers highly accurate and stable data when properly applied.


The inputs to the Model 35B are protected against transient overvoltages of up to 120 volts or 65 volts continuous.

Four Model 35Bs can be used with one Model 134 or Model 140 A/D card. Two Model 35Bs can be used with one Model 141 A/D card. Systems up to 512 channels can be assembled using Model 66 Expansion modules. Much larger systems are possible. Contact the factory for details.

may want to connect one source only and then check its operation before connecting additional inputs.

The input connections for the amplifier are inside the case. Unplug the power supply and then remove the four screws at the extreme corners of the cover. Inside the 35B you will find 16 pairs of screw terminals and three ground terminals. For shipment, the input terminals are connected together and to ground. The ground wires prevent static discharges, which can be of extremely high voltage, from damaging the circuitry. You will find a diagram of the input terminal locations on the circuit card itself. The terminals labeled 3A are selected with a channel code of 3. Terminals labeled 3 through 14 are selected using codes 4 through 15.

FIGURE 2. MODEL 35B AMPLIFIER INPUT CONNECTIONS




Run the first input cable through one of the holes in the case. Remove the grounding wire from terminal pair #1 and connect the plus wire to the plus terminal and the minus wire to the minus terminal. If a shield is present, you may want to connect it to one of the ground terminals (labeled G). Remember that both the plus and minus inputs must be within 5 volts of ground potential for proper operation. If the voltage source is floating, then the minus input should be connected to ground. If multiple inputs each have a minus wire connected to a remote common point, then either the common point should be grounded or one (and only one) of the minus leads should be jumpered to ground at the Model 35B. See Figure 3.

CAUTION: Excessive torque can damage the ground terminals.

Connect any other sources in the same fashion. Holes are provided at the card's edge for strain relief. Secure the wires to the card with lacing cord or cable ties for protection against mechanical failure. Replace the cover and securing screws.

NOTE: If a source is removed from the input terminals, it must be replaced by a grounding wire.

FIGURE 3. EXAMPLES OF INPUT SOURCES



SECTION 3. OPERATION

After the interconnections have been completed, plug in the power cord and turn on the computer. If you are using the Lawson Labs PC64 Data Logging Software, run the program. Configure the A/D channel to which you have connected the Model 35B as a multiplexer and enter the G factor. You should now see millivolts at the amplifier input displayed on the screen. Amplifiers with gains of 100 or more are scaled to display in microvolts.

The following detailed description is for programmers who want to integrate Model 35B support into their application programs. In the following example, the Model 35B is assumed to be at A/D channel #0. The Model 134 example assumes all switches are OFF. The Model 140 example assumes switch #4 is in the ON position. The Model 141 example assumes switches #1, 4, 5 and 6 are ON and all others are OFF. If your address switches are set differently, see your A/D manual for the correct ADC address.

Enter the following program. Set BOARDCODE in line 2030 to 128 for a Model 141; set it to 0 for a Model 134 or 140. If you are using a Model 141, refer to the A/D board manual and be sure to calibrate the 141 card in bipolar mode before proceeding.

The main program below begins at line 1000. It calls various subroutines to perform the necessary operations. Subroutine 2000 sets the constants for your particular combination of Model 35B and A/D board. PAUSE for the Model 141 should be set to the settling time of the A/D board plus 0.1 second. The settling time is listed in the 141 manual. The 35B itself requires only one millisecond settling time. The 0.1 second pause is needed because of the resolution of the TIMER function. Subroutine 3000 reads and displays the offset from channel #0 of the Model 35B. The format for display may be changed if desired to display fewer digits. Subroutine 7000 provides the necessary delay for the

hardware. Subroutine 8000, which depends on A/D board type, does the actual reading of the data and the conversion to millivolts.

```
1000 REM Sample program
1010 GOSUB 2000: REM Set constants for your system
1020 ADCHAN = 0: REM Assume A/D board channel #0
1030 GOSUB 3000: REM Read and display offset
1090 END

2000 REM Set constants
2010 PAUSE = 0.1: REM For the 141, enter settling time + 0.1
2020 ADC = [value]: REM Your A/D board address
2030 BOARDCODE = [value]: REM 128 for Model 141, 0 for others
2070 RETURN

3000 REM Read and display offset
3010 OUT ADC, BOARDCODE + ADCHAN * 16
3020 GOSUB 8000
3030 OFFSET = V
3040 PRINT USING "A/D Channel ##"; ADCHAN;
3050 PRINT USING "      Offset #####.#### mv"; OFFSET
3060 RETURN

7000 REM Delay for settling time of PAUSE sec.
7010 IF TIMER > 86395 THEN GOTO 7010
7020 START = TIMER
7030 IF TIMER < START + PAUSE THEN GOTO 7030
7040 RETURN
```

Include this subroutine if you are using a Model 140.

```
8000 REM Returns Model 140 voltage in millivolts
8010 OUT ADC + 1, 0
8020 IF INP(ADC) AND 128 THEN 8020
8030 X = INP(ADC)
8040 V = X MOD 16 + (INP(ADC + 1) MOD 16) * 10 + (INP(ADC + 2)
      MOD 16) * 100 + (INP(ADC + 3) MOD 16) * 1000
8050 IF X AND 32 THEN V = V + 10000
8060 IF X AND 64 THEN V = -V
8070 V = V * .25
8080 RETURN
```

Include this subroutine if you are using a Model 134. In line 8030, you may set AVERAGE anywhere from 1 to 255. Larger numbers reduce noise while smaller numbers increase speed.

```
8000 REM Returns Model 134 voltage in millivolts
8010 GOSUB 7000
8020 X = INP(ADC)
8030 TOTAL = 0: AVERAGE = 20
8040 FOR X = 1 TO AVERAGE
8050 IF INP(ADC - 1) AND 2 THEN 8050
8060 V = INP(ADC + 1) * 256! + INP(ADC)
8070 TOTAL = TOTAL + V
8080 NEXT X
8090 TOTAL = TOTAL / AVERAGE: V = (TOTAL - 32768) * .152588
8100 RETURN
```

Include this subroutine if you are using a Model 141.

```
8000 REM Returns Model 141 voltage in millivolts
```

```

8010 GOSUB 7000
8020 OUT ADC + 1, 1
8030 H = (INP(ADC + 2) AND &H0F)
8040 M = INP(ADC + 1)
8050 L = INP(ADC)
8060 OUT ADC + 1, 0
8070 COUNT = L + M * 256 + H * 65536
8080 V = COUNT * 9.536743E-03 - 5000
8090 RETURN

```

RUN the program. The offset voltage from the 35B is displayed on the screen. The value of the offset for the amplifier should be close to zero volts. If the short-term variation exceeds a millivolt, check for loose connections or improper grounding. Note that to read the offset at channel 0 on the Model 35B, both input terminals must be connected together and to ground.

Add the following lines to your program. LASTCHAN in line 2060 is the number of millivolt sources attached to your Model 35B assuming the first source is connected to channel 1. The voltage at the input terminals will be printed at line 5090.

```

1050 FOR MUXCHAN = 1 TO LASTCHAN
1060 GOSUB 5000: REM Read and display input
1070 NEXT MUXCHAN

2050 G = [value]: REM from Model 35B case
2060 LASTCHAN = [value]: REM number of sources attached


5000 REM Read and display input
5010 M35BCHAN = MUXCHAN
5030 OUT ADC, BOARDCODE + ADCHAN * 16 + M35BCHAN
5040 GOSUB 8000
5050 V = (V - OFFSET) * G / 100
5070 PRINT USING "A/D Channel ##"; ADCHAN;
5080 PRINT USING "    35B Channel ##"; M35BCHAN;
5090 PRINT USING "    ###.##### mv"; V
5100 RETURN

```

SECTION 4. USING MULTIPLE MODEL 35Bs

It is possible to connect two, three, or four Model 35Bs. When multiple Model 35Bs are connected, all must be powered for the system to function properly.

FIGURE 4. INTERCONNECTION DIAGRAM FOR MULTIPLE MODEL 35Bs



For multiple amplifier operation, add or change the following lines marked by asterisks in the program. MAXCHAN is the highest A/D channel number which has a Model 35B connected to it. Insert all the applicable G constants identified by their channel numbers.

```

1000 REM Sample program
1010 GOSUB 2000: REM Set constants for your system
* 1020 FOR ADCHAN = 0 TO MAXCHAN
1030 GOSUB 3000: REM Read and display offset
* 1050 FOR MUXCHAN = 1 TO LASTCHAN(ADCHAN)
1060 GOSUB 5000: REM Read and display input
1070 NEXT MUXCHAN
* 1080 NEXT ADCHAN
1090 END

2000 REM Set constants
2010 PAUSE = 0.1: REM For the 141, enter settling time + 0.1
2020 ADC = [value]: REM Your A/D board address
2030 BOARDCODE = [value]: REM 128 for Model 141, 0 for others
* 2050 G(0) = [value]: G(1) = [value]: REM etc.
* 2060 LASTCHAN(0) = [value]: LASTCHAN(1) = [value]: REM etc.
2070 RETURN

3000 REM Read and display offset
3010 OUT ADC, BOARDCODE + ADCHAN * 16
3020 GOSUB 8000
3030 OFFSET = V
3040 PRINT USING "A/D Channel ##"; ADCHAN;
3050 PRINT USING "    Offset #####.#### mv"; OFFSET
3060 RETURN

5000 REM Read and display input
5010 M35BCHAN = MUXCHAN
5030 OUT ADC, BOARDCODE + ADCHAN * 16 + M35BCHAN
5040 GOSUB 8000
* 5050 V = (V - OFFSET) * G(ADCHAN) / 100
5070 PRINT USING "A/D Channel ##"; ADCHAN;

```

```

5080 PRINT USING "      M35B Channel ##"; M35BCHAN;
5090 PRINT USING "      ###.##### mv"; V
5100 RETURN

```

RUN the program again. The voltages for all A/D channels and Model 35B channels will be displayed.

SECTION 5. TROUBLESHOOTING

GENERAL: If problems occur, first short the input terminals of the amplifier. If that channel then reads near zero volts, then the problem is probably with the source or its connections.

- A. If the voltage reads near zero when you think it shouldn't, check that the cord is plugged in and that the outlet is live. Either the amplifier input or output could be short-circuited. Double-check that guard on the A/D card is connected to GUARD on the amplifier.
- B. If the polarity is reversed, reverse the input leads.
- C. If the voltage readings seem random, check for a broken wire. Check that unused input terminals are grounded. Also make sure that the correct input channel is being selected.
- D. If the voltage readings are consistent but wrong, check for redundant signal grounds.
- E. If the readings show excessive scatter, check for loose connections. Make sure the GROUND terminal on the amplifier is connected to chassis ground at the computer and that ungrounded sources have a connection from minus in to G in the Model 35B. Perhaps shielding for the amplifier output wires is needed.
- F. If more than one device is connected to the output port, all must be turned on for any to work properly.
- G. If a particular source is reading erratically in a multiple 35B system, check the like-numbered sources on the other boxes.
- H. Erratic readings can be caused by ground loops or missing grounds. Try removing redundant ground wires or adding a connection from the minus input terminal to ground.
- I. Erratic readings can also be caused by an extremely fast computer over-running the settling ability of the Model 35B. Try adding a longer pause in the program after the channel selection but before the A/D conversion is initiated.

SECTION 6. INPUT CODE SUMMARY

CODE	BINARY				AMPLIFIER TERMINALS
	D	C	B	A	
0	0	0	0	0	INPUT #0
1	0	0	0	1	INPUT #1
2	0	0	1	0	INPUT #2
3	0	0	1	1	INPUT #3A
4	0	1	0	0	INPUT #3

5	0	1	0	1	INPUT #4
6	0	1	1	0	INPUT #5
7	0	1	1	1	INPUT #6
8	1	0	0	0	INPUT #7
9	1	0	0	1	INPUT #8
10	1	0	1	0	INPUT #9
11	1	0	1	1	INPUT #10
12	1	1	0	0	INPUT #11
13	1	1	0	1	INPUT #12
14	1	1	1	0	INPUT #13
15	1	1	1	1	INPUT #14

NOTE: Although there are input terminals on channel #0 of the Model 35B, that channel is normally used for software offset correction. To read the offset at any channel, both input terminals for that channel must be connected together and to ground.

SECTION 7. INTERNAL ADJUSTMENT & CALIBRATION

The only trim adjustment inside the Model 35B is an offset trim. It is normally set at the factory to 0.0 volts for the amplifier. The output is usually set with input code #0 selected. The offset will change slightly with temperature but the effects are cancelled in software and the drift will not cause errors. With the offset at 0.0 volts the Model 35B has approximately equal range in the positive and negative directions. If more range is needed in either direction, the offset can be moved away from zero by rotating the shaft of the trimpot with a small screwdriver.

There is no hardware gain adjustment. If you want to fine tune the gain, adjust the G factor to match a known input voltage. If you wish to have the calibration checked, contact Lawson Labs for details.

SECTION 8. SPECIFICATIONS

INPUTS:	16 fully differential channels
INPUT OVERVOLTAGE PROTECTION:	120 volts transient 65 volts continuous
INPUT IMPEDANCE:	1000 megohms typical
LINEARITY:	0.05% of full scale
SETTLING TIME:	400 microseconds
DIFFERENTIAL MODE INPUT RANGE:	+/- 100 mv nominal with amplifier gain of 50
COMMON MODE INPUT RANGE:	+/- 4 volts
CONTROL INPUTS:	6 volt logic, negative true. Compatible with all Lawson Labs A/D cards.

POWER
REQUIREMENT: 105-135 VAC, 50-60HZ, 6 Watts

SIZE: 6.25 x 3.75 x 2 inches

OPTIONS: Premium low-noise amplifier
Custom gain setting (x2 to x400)
DC powered, panel mounting

LIMITED WARRANTY

All Lawson Labs, Inc. products are guaranteed against defects in materials and workmanship for a period of one year from the date of delivery. Products must be returned to Lawson Labs for warranty service. Contact Lawson Labs, Inc. at 800 321-5355 for return authorization before returning anything for service. The above warranty is in lieu of all warranties express or implied. Lawson Labs will not be liable for indirect or consequential damages caused by any defect in this product. Some states do not allow the limitation of consequential damages, so the above exclusion may not apply to you.